***************
* This script produces most of the tables and figures based on the analysis of the recruiter click data
* Author: Daniel Kopp
***************	


clear
clear matrix
clear mata
set more off	
set maxvar 8000

cap drop y x
cap set obs  10
cap gen y = 1
cap gen x = 1	
reg y x

nlcom (base0: _b[x]) (base1: _b[x]), post
estimates store base0

global workvolume_det_reg "workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6 "
global workvolume_gender_reg "workvolume_det_cat1Xgeschlecht workvolume_det_cat2Xgeschlecht workvolume_det_cat3Xgeschlecht workvolume_det_cat4Xgeschlecht workvolume_det_cat5Xgeschlecht workvolume_det_cat6Xgeschlecht"

*****************************************************************
*****************************************************************
* Part-time effect for men and women with interactions
*****************************************************************
*****************************************************************

* Part-time effect for men and women with interactions
	estimates use "$save_path\workvolume_det_interact"	
		di wordcount(e(indepvars))			// 2920 - 13 (gender, workvolume, workvolume interactions) = 2907
	estimates store workv_det_gender
				
* Robustness: Baseline estimates with full sample
	estimates use "$save_path\workvolume_det_interact_all"	
	estimates store workv_det_gender_all
	
* Robustness: With occupation FE instead of search FE
	estimates use "$save_path\workvolume_det_interact_occupFE"	
	estimates store workv_det_gender_occup	
	
* Robustness: Only candidates without skills (6 part-time categories)
	estimates use "$save_path\workvolume_det_gender_interact_no_skills"	
	estimates store workv_det_gender_nosk			
	
foreach est in 	gender gender_all gender_occup gender_nosk {
	estimates restore workv_det_`est'
		local mean_sample = e(mean_sample)
		di  e(mean_sample)
		local mean_disp = e(mean_disp)		
		local N = e(N)
		matrix b = e(b) 
		matrix b = b[1,1..13]
		matrix list b
		matrix V = e(V)
		matrix V = V[1..13,1..13]
		local obs : disp %9.0f `N'
		quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
			estadd local mean_sample   " `mean_sample' "
			estadd local mean_disp   " `mean_disp' "			
			estadd local obs  " `obs' "															
	eststo workv_det_`est'_sm
	
	estimates restore workv_det_`est'_sm
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(workvolume0: _b[geschlecht])			///
				(workvolume1: _b[geschlecht] + _b[workvolume_det_cat1] + _b[workvolume_det_cat1Xgeschlecht]) ///
				(workvolume2: _b[geschlecht] + _b[workvolume_det_cat2] + _b[workvolume_det_cat2Xgeschlecht]) ///
				(workvolume3: _b[geschlecht] + _b[workvolume_det_cat3] + _b[workvolume_det_cat3Xgeschlecht]) ///
				(workvolume4: _b[geschlecht] + _b[workvolume_det_cat4] + _b[workvolume_det_cat4Xgeschlecht]) ///
				(workvolume5: _b[geschlecht] + _b[workvolume_det_cat5] + _b[workvolume_det_cat5Xgeschlecht]) ///			
				(workvolume6: _b[geschlecht] + _b[workvolume_det_cat6] + _b[workvolume_det_cat6Xgeschlecht])  , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_`est'	
	
	estimates restore workv_det_`est'_sm
	local mean_sample = e(mean_sample)	
	local mean_disp = e(mean_disp)		
	nlcom		(workvolume0: 0)			///
				(workvolume1: _b[workvolume_det_cat1] ) ///
				(workvolume2: _b[workvolume_det_cat2] ) ///
				(workvolume3: _b[workvolume_det_cat3] ) ///
				(workvolume4: _b[workvolume_det_cat4] ) ///
				(workvolume5: _b[workvolume_det_cat5] ) ///			
				(workvolume6: _b[workvolume_det_cat6] )  , post
	estadd local mean_sample   " `mean_sample' "	
	estadd local mean_disp   " `mean_disp' "				
	estimates 	store men_`est'
	
	* Scale estimates and standard errors of contact_button_clicked estimates: 	
	foreach sex in men women {
	estimates restore `sex'_`est'
			local mean_disp = e(mean_disp)	
			local mean = e(mean_sample)	
			local mean_disp2: disp %6.3fc `mean'				
			local scale: disp %9.6g  100/`mean'
			local N = e(N)
			matrix b = e(b) * `scale'
			matrix V = e(V) * `scale' ^2
			local obs : disp %8.0f `N'
			quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
				estadd local mean_disp2   " `mean_disp2' "  
				estadd local mean_disp   " `mean_disp' "  
				estadd local obs  " `obs' "									
	eststo `sex'_`est'_sc		
	}
}	

*****************************************************************
* Baseline part-time effect for men and women with interactions in percent
*****************************************************************

	* Produce figure:
	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	coefplot 	(men_gender_sc, 	keep(workvolume0) 					nokey			color(black) msymbol(Oh) offset(+0.1)) ///
				(men_gender_sc, 	keep(`keepvar') 	label(Men)		mcolor(black)	ciopts(lcolor(black)) offset(+0.1)) ///
				(women_gender_sc,  keep(workvolume0 `keepvar') 	label(Women)	mcolor(gs10) 	ciopts(lcolor(gs10)) offset(-0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(workvolume0 workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6) ///
				coeflabels(workvolume0 = "Full-time (Ref.-Cat.)" workvolume1 = "90%-99%" workvolume2 = "80%-89%" workvolume3 = "70%-79%" workvolume4 = "60%-69%" workvolume5 = "50%-59%" workvolume6 = "<50%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)") ///
				xlabel(-30(10)0)
	graph export "$results_part_time\figure_2_b.eps", as(eps) replace	
	
	* Produce table to have numerical estimates
	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	esttab men_gender_sc women_gender_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume0 `keepvar') ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab men_gender_sc women_gender_sc using "$results_part_time\figure_2_b_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01) ///
		keep(workvolume0 `keepvar') ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		

	
*****************************************************************
* Robustness: Baseline estimates with full sample	
*****************************************************************	

	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	coefplot 	(men_gender_all_sc, 	keep(workvolume0) 					nokey			color(black) msymbol(Oh) offset(+0.1)) ///
				(men_gender_all_sc, 	keep(`keepvar') 	label(Men)		mcolor(black)	ciopts(lcolor(black)) offset(+0.1)) ///
				(women_gender_all_sc,  keep(workvolume0 `keepvar') 	label(Women)	mcolor(gs10) 	ciopts(lcolor(gs10)) offset(-0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(workvolume0 workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6) ///
				coeflabels(workvolume0 = "Full-time (Ref.-Cat.)" workvolume1 = "90%-99%" workvolume2 = "80%-89%" workvolume3 = "70%-79%" workvolume4 = "60%-69%" workvolume5 = "50%-59%" workvolume6 = "<50%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)") 
	graph export "$results_part_time\figure_d9_a.eps", as(eps) replace	

	* Produce table to have numerical estimates
	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	esttab men_gender_all_sc women_gender_all_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab men_gender_all_sc women_gender_all_sc using "$results_part_time\figure_d9_a_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01) ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	
	
*****************************************************************
* Robustness: With occupation FE instead of search FE
*****************************************************************	

	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	coefplot 	(men_gender_occup_sc, 	 keep(workvolume0) 					nokey			color(black) msymbol(Oh) offset(+0.1)) ///
				(men_gender_occup_sc, 	 keep(`keepvar') 	label(Men)		mcolor(black)	ciopts(lcolor(black)) offset(+0.1)) ///
				(women_gender_occup_sc,  keep(workvolume0 `keepvar') 	label(Women)	mcolor(gs10) 	ciopts(lcolor(gs10)) offset(-0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(workvolume0 workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6) ///
				coeflabels(workvolume0 = "Full-time (Ref.-Cat.)" workvolume1 = "90%-99%" workvolume2 = "80%-89%" workvolume3 = "70%-79%" workvolume4 = "60%-69%" workvolume5 = "50%-59%" workvolume6 = "<50%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)") 
	graph export "$results_part_time\figure_d9_c.eps", as(eps) replace		
	
	* Produce table to have numerical estimates
	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	esttab men_gender_occup_sc women_gender_occup_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab men_gender_occup_sc women_gender_occup_sc using "$results_part_time\figure_d9_c_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01) ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	
		
*****************************************************************
* Robustness: Only candidates without skills	
*****************************************************************		

	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	coefplot 	(men_gender_nosk_sc, 	 keep(workvolume0) 					nokey			color(black) msymbol(Oh) offset(+0.1)) ///
				(men_gender_nosk_sc, 	 keep(`keepvar') 	label(Men)		mcolor(black)	ciopts(lcolor(black)) offset(+0.1)) ///
				(women_gender_nosk_sc,  keep(workvolume0 `keepvar') 	label(Women)	mcolor(gs10) 	ciopts(lcolor(gs10)) offset(-0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(workvolume0 workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6) ///
				coeflabels(workvolume0 = "Full-time (Ref.-Cat.)" workvolume1 = "90%-99%" workvolume2 = "80%-89%" workvolume3 = "70%-79%" workvolume4 = "60%-69%" workvolume5 = "50%-59%" workvolume6 = "<50%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)") 
	graph export "$results_part_time\figure_d9_b.eps", as(eps) replace	
	
	* Produce table to have numerical estimates
	local keepvar " workvolume1 workvolume2 workvolume3 workvolume4 workvolume5 workvolume6"
	esttab men_gender_nosk_sc women_gender_nosk_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab men_gender_nosk_sc women_gender_nosk_sc using "$results_part_time\figure_d9_b_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01) ///
		keep(workvolume0 `keepvar') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		order(workvolume0  "Full-time (Ref.)" workvolume1  "90%-99%" workvolume2  "80%-89%" workvolume3  "70%-79%" workvolume4  "60%-69%" workvolume5  "50%-59%" workvolume6  "<50%") ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
		
		
*****************************************************************
* Overall part-time effect for men and women (no part-time categories)	
* No figure or table in the article but numbers reported in text	
*****************************************************************		
	
	estimates use "$save_path\workvolume_interact_part_dummy_gender"	
	estimates store parttime_gender
	
	estimates restore parttime_gender
		local mean_sample = e(mean_sample)
		di  e(mean_sample)
		local mean_disp = e(mean_disp)		
		local N = e(N)
		matrix b = e(b) 
		matrix b = b[1,1..3]
		matrix list b
		matrix V = e(V)
		matrix V = V[1..3,1..3]
		local obs : disp %9.0f `N'
		quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
			estadd local mean_sample   " `mean_sample' "
			estadd local mean_disp   " `mean_disp' "			
			estadd local obs  " `obs' "															
	eststo parttime_gender_sm
	
	estimates restore parttime_gender_sm
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(base: 		_b[geschlecht])			///
				(parttime: 	_b[geschlecht] + _b[parttime] + _b[parttimeXgeschlecht] ), post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_parttime_gender
	
	estimates restore parttime_gender_sm
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(base: 		0)			///
				(parttime: 	_b[parttime] ), post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store men_parttime_gender
	
	* Scale estimates and standard errors of contact_button_clicked estimates: 	
	foreach sex in men women {
	estimates restore `sex'_parttime_gender
			local mean_disp = e(mean_disp)
			local mean = e(mean_sample)	
			local scale: disp %9.6g  100/`mean'
			local N = e(N)
			matrix b = e(b) * `scale'
			matrix V = e(V) * `scale' ^2
			local obs : disp %8.0f `N'
			quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
				estadd local mean_disp   " `mean_disp' "  
				estadd local obs  " `obs' "									
	eststo `sex'_parttime_gender_sc		
	}	

	esttab men_parttime_gender_sc women_parttime_gender_sc, ///
		varwidth(30) cells("b(star fmt(%6.1fc))" "se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( base parttime ) ///
		refcat(base "Full-time men" ) ///
		coeflabels(base "Full-time women" parttime "Part-time" ) ///		
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		

* No figure or table in the article but numbers reported in the text	
		
*****************************************************************	
*****************************************************************
* Only Part-time effect (without gender interaction)
*****************************************************************	
*****************************************************************

* Only det. part-time effect with gender interaction
	estimates use "$save_path\workvolume_det_only"	
	estimates store work_det_only
		
* For Placebo: Click on profile - only those without skills
	estimates use "$save_path\click_profile_placebo_no_interact"								
	estimates store work_det_clickprof_plac		
	
* Only det. part-time effect without interactions - in order to compare workvolume effect with other effects	
	estimates use "$save_path\only_workvolume_det_no_interact"								
	estimates store work_det_nolasso		
	
foreach est in 	only  clickprof_plac nolasso {
* Scale estimates and standard errors of contact_button_clicked estimates: 	
	estimates restore work_det_`est'
		local mean_disp = e(mean_disp)
		local mean = e(mean_sample)	
		local scale: disp %9.6g  100/`mean'
		local N = e(N)
		matrix b = e(b) * `scale'
		matrix V = e(V) * `scale' ^2
		local obs : disp %8.0f `N'
		quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
			estadd local mean_disp   " `mean_disp' "  
			estadd local obs  " `obs' "									
	eststo work_det_`est'_sc			
}
	
	* Overall effect of part-time  
	local keepvar " workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6"
	coefplot 	(base0, 	keep(base0) 		nokey			color(black) msymbol(Oh) offset(+0.1)) ///
				(work_det_only_sc, 	keep(`keepvar') 	nokey mcolor(black)	ciopts(lcolor(black)) offset(+0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(base0 workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6) ///
				coeflabels(base0 = "Full-time (Ref.-Cat.)" workvolume_det_cat1 = "90%-99%" workvolume_det_cat2 = "80%-89%" workvolume_det_cat3 = "70%-79%" workvolume_det_cat4 = "60%-69%" workvolume_det_cat5 = "50%-59%" workvolume_det_cat6 = "<50%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)")  ///
				xlabel(-30(10)0)
	graph export "$results_part_time\figure_2_a.eps", as(eps) replace	

	* We exclude the gender dummy for the table 
	estimates restore work_det_only	
		local mean_disp = e(mean_disp)
		local N = e(N)
		matrix b = e(b) 
		matrix b = b[1,2..7]
		matrix list b
		matrix V = e(V)
		matrix V = V[2..7,2..7]
		local obs : disp %9.0f `N'		
	quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
				estadd local mean_disp   " `mean_disp' "  
				estadd local obs  " `obs' "		
	eststo 	work_det_only_nofem		
					
	* Produce table to have numerical estimates
	estimates restore work_det_only
	local mean_sample = e(mean_sample)
	local scale_factor = 100/`mean_sample'
	disp "`scale_factor'"
	esttab work_det_only_nofem workv_det_gender_sm , ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6 geschlecht  workvolume_det_cat1Xgeschlecht workvolume_det_cat2Xgeschlecht workvolume_det_cat3Xgeschlecht workvolume_det_cat4Xgeschlecht workvolume_det_cat5Xgeschlecht workvolume_det_cat6Xgeschlecht) ///
		transform(`scale_factor'*@ `scale_factor') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume_det_cat1 "90\%-99\%" workvolume_det_cat2 "80\%-89\%" workvolume_det_cat3 "70\%-79\%" workvolume_det_cat4 "60\%-69\%" workvolume_det_cat5 "50\%-59\%" workvolume_det_cat6 "$<$50\%" geschlecht  "Female"  workvolume_det_cat1Xgeschlecht  "Female \# 90\%-99\%"  workvolume_det_cat2Xgeschlecht "Female \# 80\%-89\%" workvolume_det_cat3Xgeschlecht "Female \# 70\%-79\%" workvolume_det_cat4Xgeschlecht "Female \# 60\%-69\%" workvolume_det_cat5Xgeschlecht "Female \# 50\%-59\%"  workvolume_det_cat6Xgeschlecht "Female \# $<$50\%" ) ///		
		order(workvolume_det_cat1 "90\%-99\%" workvolume_det_cat2 "80\%-89\%" workvolume_det_cat3 "70\%-79\%" workvolume_det_cat4 "60\%-69\%" workvolume_det_cat5 "50\%-59\%" workvolume_det_cat6 "$<$50\%" geschlecht  "Female"  workvolume_det_cat1Xgeschlecht  "Female \# 90\%-99\%"  workvolume_det_cat2Xgeschlecht "Female \# 80\%-89\%" workvolume_det_cat3Xgeschlecht "Female \# 70\%-79\%" workvolume_det_cat4Xgeschlecht "Female \# 60\%-69\%" workvolume_det_cat5Xgeschlecht "Female \# 50\%-59\%"  workvolume_det_cat6Xgeschlecht "Female \# $<$50\%" ) ///		
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab work_det_only_nofem workv_det_gender_sm using "$results_part_time\figure_2_a_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6 geschlecht  workvolume_det_cat1Xgeschlecht workvolume_det_cat2Xgeschlecht workvolume_det_cat3Xgeschlecht workvolume_det_cat4Xgeschlecht workvolume_det_cat5Xgeschlecht workvolume_det_cat6Xgeschlecht) ///
		transform(`scale_factor'*@ `scale_factor') ///
		refcat(workvolume_det_cat1 "Full-time" ) ///
		coeflabels(workvolume_det_cat1 "90\%-99\%" workvolume_det_cat2 "80\%-89\%" workvolume_det_cat3 "70\%-79\%" workvolume_det_cat4 "60\%-69\%" workvolume_det_cat5 "50\%-59\%" workvolume_det_cat6 "$<$50\%" geschlecht  "Female"  workvolume_det_cat1Xgeschlecht  "Female \# 90\%-99\%"  workvolume_det_cat2Xgeschlecht "Female \# 80\%-89\%" workvolume_det_cat3Xgeschlecht "Female \# 70\%-79\%" workvolume_det_cat4Xgeschlecht "Female \# 60\%-69\%" workvolume_det_cat5Xgeschlecht "Female \# 50\%-59\%"  workvolume_det_cat6Xgeschlecht "Female \# $<$50\%" ) ///		
		order(workvolume_det_cat1 "90\%-99\%" workvolume_det_cat2 "80\%-89\%" workvolume_det_cat3 "70\%-79\%" workvolume_det_cat4 "60\%-69\%" workvolume_det_cat5 "50\%-59\%" workvolume_det_cat6 "$<$50\%" geschlecht  "Female"  workvolume_det_cat1Xgeschlecht  "Female \# 90\%-99\%"  workvolume_det_cat2Xgeschlecht "Female \# 80\%-89\%" workvolume_det_cat3Xgeschlecht "Female \# 70\%-79\%" workvolume_det_cat4Xgeschlecht "Female \# 60\%-69\%" workvolume_det_cat5Xgeschlecht "Female \# 50\%-59\%"  workvolume_det_cat6Xgeschlecht "Female \# $<$50\%" ) ///		
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		

	
*****************************************************************
* Plabebo estimates
*****************************************************************	

	* Age and last wage effects for full sample	
	coefplot 	(base0 			, 			keep(base1 ) nokey nokey		color(black) msymbol(Oh) offset(-0.1)) ///	
				(work_det_only_sc, 			keep( alter_cat1 base2 alter_cat3 alter_cat4 alter_cat5  versich_verdienst  ) nokey 	color(black)	ciopts(lcolor(black)) offset(-0.1)	) , ///
				order(  versich_verdienst alter_cat1 base1 alter_cat3 alter_cat4 alter_cat5  ) ///
				xtitle("Effect (in %) on contact likelihood" , margin(medsmall) )   ///
				xlabels(-2(1)2) ///
				coeflabels( alter_cat1 = "<30 years" base1 = "30-39 years (Ref.-Cat.)" 	alter_cat3 = "40-49 years" alter_cat4 = "50-60 years" alter_cat5 = "> 60 years" versich_verdienst = "Last wage (in 1'000)" ) ///
				headings( alter_cat1 = "{bf: Age}" ) /// 
				graphregion(color(white)) bgcolor(white) omitted xline(0, lcolor(gs10) lpattern(dash)) 				
	graph export "$results_part_time\figure_b1_a.eps", as(eps) replace	
	
	* Produce table to have numerical estimates
	esttab work_det_only_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( alter_cat1  alter_cat3 alter_cat4 alter_cat5  versich_verdienst  )  ///
		coeflabels( alter_cat1  "<30 years" 	alter_cat3  "40-49 years" alter_cat4  "50-60 years" alter_cat5  "> 60 years" versich_verdienst  "Last wage (in 1'000)" ) ///
		refcat(alter_cat3 "30-39 years (Ref.-Cat.)" ) ///
		title(Effects of characteristics not visible to recruiters on Job-Room) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab work_det_only_sc using "$results_part_time\figure_b1_a_table.csv", replace /// 
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( alter_cat1  alter_cat3 alter_cat4 alter_cat5  versich_verdienst  )  ///
		coeflabels( alter_cat1  "<30 years" 	alter_cat3  "40-49 years" alter_cat4  "50-60 years" alter_cat5  "> 60 years" versich_verdienst  "Last wage (in 1'000)" ) ///
		refcat(alter_cat3 "30-39 years (Ref.-Cat.)" ) ///
		title(Effects of characteristics not visible to recruiters on Job-Room) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	
	* Click on profile 	
	cap gen y = 1 
	cap gen base1 = 1
	reg y base1
	est store base	

	estimates restore base
	nlcom (base4: _b[base1]) (base5:  _b[base1])  (base6:  _b[base1])  (base7:  _b[base1])  (base8:  _b[base1]) (base9:  _b[base1]) , post
	estimates store base_cat
	
	* Note that placebo estimates with click on profile as dep variable contain only profiles without skills
	coefplot 	(base_cat,					keep(base4  base6 ) nokey  msymbol(circle_hollow) color(black)  ) ///
				(work_det_clickprof_plac_sc, keep(abschluss_s2 abschluss_s3         erfa_s1 erfa_s2 erfa_s3          )  nokey  color(black) ciopts(lcolor(black)) offset(+0.2)) , ///
				order(   base4 abschluss_s2 abschluss_s3  base5  base6 erfa_s1 erfa_s2 erfa_s3             ) ///
				coeflabels(  base4 = "Swiss (Ref)" abschluss_s2 = "Foreign in CH accepted" abschluss_s3 = "Foreign in CH not accepted"    base6 = "No experience (Ref)"  erfa_s1 = "< 1 year exp." erfa_s2 = "1-3 years exp." erfa_s3 = "At least 4 years exp."   geschlecht = "Female"  gesch_kontakt = "Contact information visible" verfuegbar1 = "Immediately available"  l_German = "Very goog knowledge of German" l_CH_German = "Very good knowledge of Swiss German" l_English = "Very good knowledge of English" l_French = "Very good knowledge of French" l_Italian = "Very good knowledge of Italian" has_skills = "Skills reported" sk_softskills = "Softskills" sk_it_gen = "General IT skills" sk_it_deep = "Special IT skills"  ) ///
				xtitle(" " "Effect on profile click likelihood (in %)" )   ///
				xlabel(-2(1)2) ///
				graphregion(color(white)) bgcolor(white) omitted xline(0, lcolor(gs10) lp(dash)) rescale(`scale_factor') ///    
				headings(base4 = "{bf:Education certificate}" base6 = "{bf:Experience}"   ) 	
	graph export "$results_part_time\figure_b1_b.eps", as(eps) replace

	* Produce table to have numerical estimates
	esttab work_det_clickprof_plac_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(abschluss_s2 abschluss_s3         erfa_s1 erfa_s2 erfa_s3          )   ///
		coeflabels( abschluss_s2  "Foreign in CH accepted" abschluss_s3  "Foreign in CH not accepted"     erfa_s1  "< 1 year exp." erfa_s2  "1-3 years exp." erfa_s3 "At least 4 years exp."   ) ///
		refcat(abschluss_s2 "Swiss (Ref)" erfa_s1 "No experience (Ref)") ///
		title(Effects of characteristics not visible to recruiters on Job-Room) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab work_det_clickprof_plac_sc using "$results_part_time\figure_b1_b_table.csv", replace /// 
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(abschluss_s2 abschluss_s3         erfa_s1 erfa_s2 erfa_s3          )   ///
		coeflabels( abschluss_s2  "Foreign in CH accepted" abschluss_s3  "Foreign in CH not accepted"     erfa_s1  "< 1 year exp." erfa_s2  "1-3 years exp." erfa_s3  "At least 4 years exp."   ) ///
		refcat(abschluss_s2 "Swiss (Ref)" erfa_s1 "No experience (Ref)") ///
		title(Effects of characteristics not visible to recruiters on Job-Room) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	
	
	
*****************************************************************
* Effect of different job characteristics on contact likelihood
*****************************************************************	
	
	cap gen y = 1 
	cap gen base1 = 1
	reg y base1
	est store base	

	estimates restore base
	nlcom (base4: _b[base1]) (base5:  _b[base1])  (base6:  _b[base1])  (base7:  _b[base1])  (base8:  _b[base1]) (base9:  _b[base1]) , post
	estimates store base_cat

	coefplot 	(base_cat,				keep(base4 base5 base6 base7) nokey  msymbol(circle_hollow) color(black) offset(-0.05) ) ///
				(work_det_nolasso_sc,	keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6  abschluss_s2 abschluss_s3 cod_ausbildungsniveau132 cod_ausbildungsniveau134 cod_ausbildungsniveau171 cod_ausbildungsniveau173     erfa_s1 erfa_s2 erfa_s3  l_German l_CH_German    sk_softskills sk_it_gen sk_it_deep     )  nokey color(black) ciopts(lcolor(black)) ) , ///
				order(base7 workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6 base4 abschluss_s2 abschluss_s3  base5 cod_ausbildungsniveau132 cod_ausbildungsniveau134 cod_ausbildungsniveau171 cod_ausbildungsniveau173 base6 erfa_s1 erfa_s2 erfa_s3   l_German l_CH_German    sk_softskills sk_it_gen sk_it_deep      ) ///
				coeflabels(base7 = "Fulltime (Ref)" workvolume_det_cat1 = "90%-99%" workvolume_det_cat2 = "80%-89%" workvolume_det_cat3 = "70%-79%" workvolume_det_cat4 = "60%-69%" workvolume_det_cat5 = "50%-59%" workvolume_det_cat6 = "<50%" base4 = "Swiss (Ref)" abschluss_s2 = "Foreign in CH accepted" abschluss_s3 = "Foreign in CH not accepted" base5 =  "Compulsory school (Ref)"  cod_ausbildungsniveau132 = "Secondary vocational education (EFZ)" cod_ausbildungsniveau134 = "Higher vocational diploma (Berfusmaturität)"		cod_ausbildungsniveau171 = "Bachelor or equivalent"	 cod_ausbildungsniveau173 = "Master or equivalent"   base6 = "No experience (Ref)"  erfa_s1 = "< 1 year exp." erfa_s2 = "1-3 years exp." erfa_s3 = "At least 4 years exp."   geschlecht = "Female"  gesch_kontakt = "Contact information visible" verfuegbar1 = "Immediately available"  l_German = "Very good knowledge of German" l_CH_German = "Very good knowledge of Swiss German" l_English = "Very good knowledge of English" l_French = "Very good knowledge of French" l_Italian = "Very good knowledge of Italian" has_skills = "Skills reported" sk_softskills = "Softskills" sk_it_gen = "General IT skills" sk_it_deep = "Special IT skills"  , labsize(vsmall) ) ///
				xtitle(" " "Effect on contact likelihood (in %)" )   ///
				xlabel(-25(5)15) ///
				graphregion(color(white)) bgcolor(white) omitted xline(0, lcolor(gs10) lp(dash)) rescale(`scale_factor') ///    
				headings(base7 = "{bf: Full-/Part-time}" base4 = "{bf:Education certificate}" base5 = "{bf:Education}" base6 = "{bf:Experience}" 1.geschlecht = "{bf:Other Characteristics}" l_German = "{bf:Language skills}" sk_softskills = "{bf:Other skills (selection)}" ) 	
	graph export "$results_part_time\figure_d4.eps", as(eps) replace	
	
	esttab work_det_nolasso_sc, keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6  abschluss_s2 abschluss_s3 cod_ausbildungsniveau132 cod_ausbildungsniveau134 cod_ausbildungsniveau171 cod_ausbildungsniveau173     erfa_s1 erfa_s2 erfa_s3  l_German l_CH_German    sk_softskills sk_it_gen sk_it_deep  geschlecht  ) ///
					varwidth(30) 
	
	* Produce table to have numerical estimates
	esttab work_det_nolasso_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6  abschluss_s2 abschluss_s3 cod_ausbildungsniveau132 cod_ausbildungsniveau134 cod_ausbildungsniveau171 cod_ausbildungsniveau173     erfa_s1 erfa_s2 erfa_s3  l_German l_CH_German    sk_softskills sk_it_gen sk_it_deep          )   ///
		coeflabels(workvolume_det_cat1  "90%-99%" workvolume_det_cat2  "80%-89%" workvolume_det_cat3  "70%-79%" workvolume_det_cat4 "60%-69%" workvolume_det_cat5 "50%-59%" workvolume_det_cat6 "<50%"  abschluss_s2 "Foreign in CH accepted" abschluss_s3 "Foreign in CH not accepted"   cod_ausbildungsniveau132 "Secondary vocational education (EFZ)" cod_ausbildungsniveau134  "Higher vocational diploma (Berfusmaturität)"		cod_ausbildungsniveau171 "Bachelor or equivalent"	 cod_ausbildungsniveau173 "Master or equivalent"    erfa_s1  "< 1 year exp." erfa_s2  "1-3 years exp." erfa_s3  "At least 4 years exp."   l_German  "Very good knowledge of German" l_CH_German  "Very good knowledge of Swiss German" l_English  "Very good knowledge of English" l_French  "Very good knowledge of French" l_Italian  "Very good knowledge of Italian" sk_softskills  "Softskills" sk_it_gen  "General IT skills" sk_it_deep  "Special IT skills"   ) ///
		refcat(workvolume_det_cat1 "Fulltime (Ref)" abschluss_s2 "Swiss (Ref)"  cod_ausbildungsniveau132 "Compulsory school (Ref)" erfa_s1 "No experience (Ref)"  ) ///
		title(Effect on contact likelihood (in %)) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab work_det_clickprof_plac_sc using "$results_part_time\figure_d4_table.csv", replace /// 
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(workvolume_det_cat1 workvolume_det_cat2 workvolume_det_cat3 workvolume_det_cat4 workvolume_det_cat5 workvolume_det_cat6  abschluss_s2 abschluss_s3 cod_ausbildungsniveau132 cod_ausbildungsniveau134 cod_ausbildungsniveau171 cod_ausbildungsniveau173     erfa_s1 erfa_s2 erfa_s3  l_German l_CH_German    sk_softskills sk_it_gen sk_it_deep          )   ///
		coeflabels(workvolume_det_cat1  "90%-99%" workvolume_det_cat2  "80%-89%" workvolume_det_cat3  "70%-79%" workvolume_det_cat4 "60%-69%" workvolume_det_cat5 "50%-59%" workvolume_det_cat6 "<50%"  abschluss_s2 "Foreign in CH accepted" abschluss_s3 "Foreign in CH not accepted"   cod_ausbildungsniveau132 "Secondary vocational education (EFZ)" cod_ausbildungsniveau134  "Higher vocational diploma (Berfusmaturität)"		cod_ausbildungsniveau171 "Bachelor or equivalent"	 cod_ausbildungsniveau173 "Master or equivalent"    erfa_s1  "< 1 year exp." erfa_s2  "1-3 years exp." erfa_s3  "At least 4 years exp."   l_German  "Very good knowledge of German" l_CH_German  "Very good knowledge of Swiss German" l_English  "Very good knowledge of English" l_French  "Very good knowledge of French" l_Italian  "Very good knowledge of Italian" sk_softskills  "Softskills" sk_it_gen  "General IT skills" sk_it_deep  "Special IT skills"   ) ///
		refcat(workvolume_det_cat1 "Fulltime (Ref)" abschluss_s2 "Swiss (Ref)"  cod_ausbildungsniveau132 "Compulsory school (Ref)" erfa_s1 "No experience (Ref)"  ) ///
		title(Effects of characteristics not visible to recruiters on Job-Room) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	
	
	
*****************************************************************
* Effect of being a women in searches in which recruiters restrict results to full- or part-time jobseekers
*****************************************************************

estimates use "$save_path\gender_new_s_noworkload"
disp e(mean_sample)
estimates store gender_s_noworkload

estimates use "$save_path\gender_new_s_part"
disp e(mean_sample)
estimates store gender_s_part

estimates use "$save_path\gender_new_s_full"
disp e(mean_sample)
estimates store gender_s_full

estimates use "$save_path\gender_new_s_noworkload_part"
disp e(mean_sample)
estimates store gender_s_noworkload_p


	foreach est in s_part s_full s_noworkload s_noworkload_p {
	estimates restore gender_`est'
			local mean_disp = e(mean_disp)
			local mean = e(mean_sample)	
			local scale: disp %9.6g  100/`mean'
			local N = e(N)
			matrix b = e(b) * `scale'
			matrix V = e(V) * `scale' ^2
			local obs : disp %8.0f `N'
			quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
				estadd local mean_disp   " `mean_disp' "  
				estadd local obs  " `obs' "									
	eststo gender_`est'_sc		
	}	

	esttab gender_s_noworkload_p_sc gender_s_noworkload_sc gender_s_part_sc gender_s_full_sc, ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar collabels(none)  star(* 0.1 ** 0.05 *** 0.01)  ///
		keep(geschlecht parttime parttimeXgeschlecht) ///
		refcat(geschlecht "Men") ///
		coeflabels(geschlecht "Women"   parttime "Part-time"  parttimeXgeschlecht "Part-time*Women") ///		
		mtitle("Not restricted" "Not restricted" "Only part-time" "Only full-time") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
		
	esttab gender_s_noworkload_p_sc gender_s_noworkload_sc gender_s_part_sc gender_s_full_sc using "$results_part_time/table_d6.tex", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar collabels(none)  star(* 0.1 ** 0.05 *** 0.01) frag ///
		keep(geschlecht parttime parttimeXgeschlecht) ///
		refcat(geschlecht "Men") ///
		coeflabels(geschlecht "Women"   parttime "Part-time"  parttimeXgeschlecht "Part-time*Women") ///		
		mtitle("Not restricted" "Not restricted" "Only part-time" "Only full-time") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")			
	
	
	
***************************************************************************************
* Effect of workvolume and gender by ISCO 1 occupations
**************************************************************************************
	
	********************************************************************	
	* The effect of part-time work and gender by isco 1
	********************************************************************
		
	estimates use "$save_path\parttime_gender_by_isco1_int"
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)		
	local N = e(N)
	matrix b = e(b) 
	matrix b = b[1,1..30]
	matrix V = e(V)
	matrix V = V[1..30,1..30]
	local obs : disp %9.0f `N'
	quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
		estadd local mean_sample   " `mean_sample' "
		estadd local mean_disp   " `mean_disp' "			
		estadd local obs  " `obs' "															
	eststo part_isco1_int_s
		
	* Ratio female to male part-time penalty
	forvalues i = 0/9 {
	estimates restore part_isco1_int_s

	estimates restore part_isco1_int_s
	local mean_sample = e(mean_sample)	
	local mean_disp = e(mean_disp)		
	nlcom		(fulltime: 0)			///
				(parttime: (_b[isco1_`i'Xparttime] + _b[isco1_`i'XparttimeXgeschlecht])/_b[isco1_`i'Xparttime] ) , post
	estadd local mean_sample   " `mean_sample' "	
	estadd local mean_disp   " `mean_disp' "				
	estimates 	store ratio_p_isco1_`i'		
	}
			
	cap label define label_isco_1_short 0 "Unknown" 1 "Managers" 2 "Professionals" 3 "Technicians" 4 "Clerks" 5 "Service/Sales" 6 "Agriculture" 7 "Craft" 8 "Machine operators" 9 "Elementary"	
		
	* Ratio female to male part-time penalty
		local color1 "black"
		local keep_vars "parttime "
		coefplot (ratio_p_isco1_0, aseq(Unknown)   	keep(`keep_vars')   ///
			\ ratio_p_isco1_1, aseq(Managers)  		keep(`keep_vars')  ///
			\ ratio_p_isco1_2, aseq(Professionals)  keep(`keep_vars')  ///
			\ ratio_p_isco1_3, aseq(Technicians)  	keep(`keep_vars')  ///
			\ ratio_p_isco1_4, aseq(Clerks)  		keep(`keep_vars')  ///
			\ ratio_p_isco1_5, aseq(Service/Sales) 	keep(`keep_vars') /// 
			\ ratio_p_isco1_6, aseq(Agriculture) 	keep(`keep_vars') ///
			\ ratio_p_isco1_7, aseq(Craft) 			keep(`keep_vars') ///
			\ ratio_p_isco1_8, aseq(Machine operators) keep(`keep_vars') ///
			\ ratio_p_isco1_9, aseq(Elementary) keep(`keep_vars') color(`color1') ciopts(lcolor(`color1')) label(Part-time)	) , /// 
			noeqlabels swapnames vertical 		///
			graphregion(color(white) margin(medium )) bgcolor(white) ///
			ytitle( "Ratio of female to male part-time penalty ", size(medsmall) ) ///
			xlabel(, labsize(small) alt) ///
			yline(1, lcolor(black) lpattern(dash)) 
		graph export "$results_part_time\figure_d7.eps", as(eps) replace		

	* Produce table to have numerical estimates
	esttab ratio_p_isco1_0 ratio_p_isco1_1 ratio_p_isco1_2 ratio_p_isco1_3 ratio_p_isco1_4 ratio_p_isco1_5 ratio_p_isco1_6 ratio_p_isco1_7 ratio_p_isco1_8 ratio_p_isco1_9, ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( parttime )   ///
		coeflabels( parttime "Ratio") ///
		refcat(workvolume_det_cat1 "Fulltime (Ref)" abschluss_s2 "Swiss (Ref)"  cod_ausbildungsniveau132 "Compulsory school (Ref)" erfa_s1 "No experience (Ref)"  ) ///
		title(Ratio of female to male part-time penalty) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab ratio_p_isco1_0 ratio_p_isco1_1 ratio_p_isco1_2 ratio_p_isco1_3 ratio_p_isco1_4 ratio_p_isco1_5 ratio_p_isco1_6 ratio_p_isco1_7 ratio_p_isco1_8 ratio_p_isco1_9 using "$results_part_time\figure_d7_table.csv", replace /// 
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( parttime )   ///
		coeflabels( parttime "Ratio") ///
		refcat(workvolume_det_cat1 "Fulltime (Ref)" abschluss_s2 "Swiss (Ref)"  cod_ausbildungsniveau132 "Compulsory school (Ref)" erfa_s1 "No experience (Ref)"  ) ///
		title(Ratio of female to male part-time penalty) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
		

	********************************
	* Part-time penalty by vote share in parental leave referendum
	********************************

	estimates use "$save_path\parttime_high_yes_share_parental_leave_int"
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)		
	local N = e(N)
	matrix b = e(b) 
	matrix b = b[1,1..6]
	matrix V = e(V)
	matrix V = V[1..6,1..6]
	local obs : disp %9.0f `N'
	quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
		estadd local mean_sample   " `mean_sample' "
		estadd local mean_disp   " `mean_disp' "			
		estadd local obs  " `obs' "															
	eststo part_vote_int_s
		
	* Part-time penalty of women in high share
	estimates restore part_vote_int_s	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXgeschlecht] + _b[parttimeXhigh_yes] + _b[parttimeXgeschlechtXhigh_yes]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_high_share	
	
	* Part-time penalty of women low share
	estimates restore part_vote_int_s
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXgeschlecht]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_low_share		
		
	* Part-time penalty of men high share
	estimates restore part_vote_int_s	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXhigh_yes]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store men_high_share	
	
	* Part-time penalty of men low share
	estimates restore part_vote_int_s	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store men_low_share		
	

	estimates restore men_high_share
	matrix b = e(b) 
	local men_high = b[1,2]
	disp "`men_high'"
	estimates restore men_low_share
	matrix b = e(b) 
	local men_low = b[1,2]
	disp "`men_low'"	
	estimates restore women_high_share
	matrix b = e(b) 
	local women_high = b[1,2]
	disp "`women_high'"
	estimates restore women_low_share
	matrix b = e(b) 
	local women_low = b[1,2]
	disp "`women_low'"			
	estimates restore part_vote_int_s
	matrix b = e(b)
	local part_geschl : disp %5.4f b[1,4]
	disp "`part_geschl'"
	estimates restore part_vote_int_s		
	nlcom		(gen_diff_high: _b[parttimeXgeschlecht] + _b[parttimeXgeschlechtXhigh_yes]) , post					
	matrix b = e(b)
	local gen_diff_high : disp %5.4f b[1,1]
	disp "`gen_diff_high'"
	estimates restore part_vote_int_s
	matrix b = e(b)
	local part_geschl_high : disp %5.4f b[1,6]
	disp "`part_geschl_high'"		
	local color1 "black"
	local color2 "black"
	local keep_vars "parttime "
	coefplot  (men_low_share, aseq("Men")  			keep(`keep_vars') color(`color2') ciopts(lcolor(`color2') lwidth(medthick)) msize(large) offset(+0.0)) ///
			( women_low_share, aseq("Women")  		keep(`keep_vars') color(`color2') ciopts(lcolor(`color2') lwidth(medthick)) msize(large) ) ///
			( men_high_share, aseq("Men ")  		keep(`keep_vars') color(`color1') ciopts(lcolor(`color1') lwidth(medthick)) msize(large) ) ///
			( women_high_share, aseq("Women ")  	keep(`keep_vars') color(`color1') ciopts(lcolor(`color1') lwidth(medthick)) msize(large) ) 	  , ///  
		noeqlabels swapnames vertical 	legend(off)	///
		graphregion(color(white) margin(b=7 )) bgcolor(white) ///
		ytitle( "{bf:Part-time penalty in PP}", size(medium) ) ///
		ylabel(0(0.01)-0.04) ///
		xlabel(, labsize(medium) ) ///
		yline(0, lcolor(black) lpattern(dash)) 		///
		text(	-0.018 4.38 "`gen_diff_high'***" 		///
				-0.027 2.38 "`part_geschl'***"			///
				, size(medsmall)  ) ///
		text(	-0.0475 1.48 "{bf:low share of yes votes}" ///
				-0.0475 3.48 "{bf:high share of yes votes}" ///					
				, size(medsmall)  ) 
		addplot: scatteri `men_high' 4.11 `women_high' 4.11	, 	recast(line) lcolor(`color1')  norescaling lwidth(medthick)
		addplot: scatteri `men_high' 4.11 `men_high' 4.08	, 		recast(line) lcolor(`color1')  norescaling  lwidth(medthick)
		addplot: scatteri `women_high' 4.11 `women_high' 4.08	, 	recast(line) lcolor(`color1')  norescaling lwidth(medthick)
		addplot: scatteri `men_low' 2.11 `women_low' 2.11, 	recast(line) lcolor(`color2')  norescaling lwidth(medthick)
		addplot: scatteri `men_low' 2.11 `men_low' 2.08	, 	recast(line) lcolor(`color2')  norescaling lwidth(medthick)
		addplot: scatteri `women_low' 2.11 `women_low' 2.08, 	recast(line) lcolor(`color2')  norescaling 	lwidth(medthick)		
	graph export "$results_part_time\figure_5_a.eps", as(eps) replace		

	* Produce table to have numerical estimates	
	esttab part_vote_int_s  ,  ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar nomtitle collabels(none)  star(* 0.1 ** 0.05 *** 0.01)  ///
		refcat(geschlecht "Men*Full-time*Low yes share") ///
		order(geschlecht geschlechtXhigh_yes parttime parttimeXhigh_yes parttimeXgeschlecht parttimeXgeschlechtXhigh_yes) ///
		coeflabels(parttime "Part-time"  parttimeXhigh_yes "Part-time*High yes share" geschlecht "Women"  geschlechtXhigh_yes "Women*High yes share"  parttimeXgeschlecht "Women*Part-time" parttimeXgeschlechtXhigh_yes "Women*Part-time*High yes share") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")	
	esttab part_vote_int_s  using "$results_part_time\figure_5_a_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar nomtitle collabels(none)  star(* 0.1 ** 0.05 *** 0.01)  ///
		refcat(geschlecht "Men*Full-time*Low yes share") ///
		order(geschlecht geschlechtXhigh_yes parttime parttimeXhigh_yes parttimeXgeschlecht parttimeXgeschlechtXhigh_yes) ///
		coeflabels(parttime "Part-time"  parttimeXhigh_yes "Part-time*High yes share" geschlecht "Women"  geschlechtXhigh_yes "Women*High yes share"  parttimeXgeschlecht "Women*Part-time" parttimeXgeschlechtXhigh_yes "Women*Part-time*High yes share") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")	


	*********************************************
	* By gender typicality of recruiter selection behavior
	*********************************************
			
	estimates use "$save_path\parttime_gender_typical_recruiter2_clicks_int"
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)		
	local N = e(N)
	matrix b = e(b) 
	matrix b = b[1,1..7]
	matrix V = e(V)
	matrix V = V[1..7,1..7]
	local obs : disp %9.0f `N'
	quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
		estadd local mean_sample   " `mean_sample' "
		estadd local mean_disp   " `mean_disp' "			
		estadd local obs  " `obs' "															
	eststo part_gendtyp2_cl
		
	* Part-time penalty of women when focusing on gender typical recruiters
	estimates restore part_gendtyp2_cl	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXgeschlecht] + _b[parttimeXgender_typ2_cl] + _b[partXgeschlechtXgender_typ2_cl ]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_gendtyp2_cl	
	
	* Part-time penalty of women when focusing on gender atypical recruiters
	estimates restore part_gendtyp2_cl
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXgeschlecht]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store women_gendatyp2_cl	
		
	* Part-time penalty of men when focusing on gender typical recruiters
	estimates restore part_gendtyp2_cl	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime] + _b[parttimeXgender_typ2_cl]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store men_gendtyp2_cl	
	
	* Part-time penalty of men when focusing on gender atypical recruiters
	estimates restore part_gendtyp2_cl	
	local mean_sample = e(mean_sample)
	local mean_disp = e(mean_disp)			
	nlcom		(fulltime: 0)			///
				(parttime: _b[parttime]) , post
	estadd local mean_sample   " `mean_sample' "
	estadd local mean_disp   " `mean_disp' "					
	estimates 	store men_gendatyp2_cl		
		
	estimates restore men_gendtyp2_cl
	matrix b = e(b) 
	local men_typ = b[1,2]
	disp "`men_typ'"
	estimates restore men_gendatyp2_cl
	matrix b = e(b) 
	local men_atyp = b[1,2]
	disp "`men_atyp'"	
	estimates restore women_gendtyp2_cl
	matrix b = e(b) 
	local women_typ = b[1,2]
	disp "`women_typ'"
	estimates restore women_gendatyp2_cl
	matrix b = e(b) 
	local women_atyp = b[1,2]
	disp "`women_atyp'"			
	estimates restore part_gendtyp2_cl
	matrix b = e(b)
	local part_geschl : disp %5.4f b[1,5]
	disp "`part_geschl'"
	estimates restore part_gendtyp2_cl		
	nlcom		(gen_diff_typ: _b[parttimeXgeschlecht] + _b[partXgeschlechtXgender_typ2_cl ]) , post					
	matrix b = e(b)
	local gen_diff_typ : disp %5.4f b[1,1]
	disp "`gen_diff_typ'"
	estimates restore part_gendtyp2_cl
	matrix b = e(b)
	local part_geschl_typ : disp %5.4f b[1,7]
	disp "`part_geschl_typ'"		
	local color1 "black"
	local color2 "black"
	local keep_vars "parttime "
	coefplot  (men_gendtyp2_cl, aseq("Men")  			keep(`keep_vars') color(`color2') ciopts(lcolor(`color2') lwidth(medthick)) msize(large) offset(+0.0)) ///
			( women_gendtyp2_cl, aseq("Women")  		keep(`keep_vars') color(`color2') ciopts(lcolor(`color2') lwidth(medthick)) msize(large) ) ///
			( men_gendatyp2_cl, aseq("Men ")  		keep(`keep_vars') color(`color1') ciopts(lcolor(`color1') lwidth(medthick)) msize(large) ) ///
			( women_gendatyp2_cl, aseq("Women ")  	keep(`keep_vars') color(`color1') ciopts(lcolor(`color1') lwidth(medthick)) msize(large) ) 	  , ///  		
		noeqlabels swapnames vertical 	legend(off)	///
		graphregion(color(white) margin(b=7 )) bgcolor(white) ///
		ytitle( "{bf:Part-time penalty in PP}", size(medium) ) ///
		ylabel(0(0.01)-0.04) ///
		xlabel(, labsize(medium) ) ///
		yline(0, lcolor(black) lpattern(dash)) 		///
		text(	-0.0183 4.38 "`part_geschl'" 		///
				-0.022 2.38 	"`gen_diff_typ '***"		///
				, size(medsmall)  ) 				///	
		text(	-0.0455 1.5 "{bf:gender-typical selection behavior}" ///
				-0.0455 3.5 "{bf:gender-atypical selection behavior}" ///
				, size(medsmall)  ) 
		addplot: scatteri `men_typ' 	2.11 `women_typ' 	2.11	, 	recast(line) lcolor(`color2')  norescaling lwidth(medthick)
		addplot: scatteri `men_typ' 	2.11 `men_typ' 		2.08	, 	recast(line) lcolor(`color2')  norescaling lwidth(medthick) 
		addplot: scatteri `women_typ' 	2.11 `women_typ' 	2.08	, 	recast(line) lcolor(`color2')  norescaling lwidth(medthick)
		addplot: scatteri `men_atyp' 	4.11 `women_atyp' 	4.11	, 	recast(line) lcolor(`color1')  norescaling lwidth(medthick)
		addplot: scatteri `men_atyp' 	4.11 `men_atyp' 	4.08	, 	recast(line) lcolor(`color1')  norescaling lwidth(medthick)
		addplot: scatteri `women_atyp' 	4.11 `women_atyp' 	4.08	, 	recast(line) lcolor(`color1')  norescaling lwidth(medthick)		
	graph export "$results_part_time\figure_5_b.eps", as(eps) replace		

	* Produce table to have numerical estimates		
	esttab part_gendtyp2_cl  ,  ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar nomtitle collabels(none)  star(* 0.1 ** 0.05 *** 0.01)  ///
		refcat(geschlecht "Men*Full-time*Gender-atypical") ///
		keep(geschlecht geschlechtXgender_typ2_cl  parttime parttimeXgender_typ2_cl  parttimeXgeschlecht partXgeschlechtXgender_typ2_cl ) ///		
		order(geschlecht geschlechtXgender_typ2_cl  parttime parttimeXgender_typ2_cl  parttimeXgeschlecht partXgeschlechtXgender_typ2_cl ) ///
		coeflabels(parttime "Part-time"  parttimeXgender_typ2_cl "Part-time*Gender-typical" geschlecht "Women"  geschlechtXgender_typ2_cl "Women*Gender-typical"  parttimeXgeschlecht "Women*Part-time" partXgeschlechtXgender_typ2_cl "Women*Part-time*Gender-typical") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")	
	esttab part_gendtyp2_cl using "$results_part_time\figure_5_b_table.csv" , replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc))" "se(par fmt(%6.2gc))") label nodepvar nomtitle collabels(none)  star(* 0.1 ** 0.05 *** 0.01)  ///
		refcat(geschlecht "Men*Full-time*Gender-atypical") ///
		keep(geschlecht geschlechtXgender_typ2_cl  parttime parttimeXgender_typ2_cl  parttimeXgeschlecht partXgeschlechtXgender_typ2_cl ) ///		
		order(geschlecht geschlechtXgender_typ2_cl  parttime parttimeXgender_typ2_cl  parttimeXgeschlecht partXgeschlechtXgender_typ2_cl ) ///
		coeflabels(parttime "Part-time"  parttimeXgender_typ2_cl "Part-time*Gender-typical" geschlecht "Women"  geschlechtXgender_typ2_cl "Women*Gender-typical"  parttimeXgeschlecht "Women*Part-time" partXgeschlechtXgender_typ2_cl "Women*Part-time*Gender-typical") ///		
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")	
	
			
		
	****************************************	
	****************************************
	* With jobseeker fixed effects 
	****************************************
	****************************************
	
	* This sample also includes searches in which recruiters enter work volume as a search criteria
	
	estimates use "$save_path\gender_perc_workvolume_4det_biglasso_stesfe_all_together"	
	disp e(N)
	estimates store stesfe_lasso	
				
	foreach fe in stes {
		foreach cov in lasso {
		estimates restore `fe'fe_`cov'
			local mean_sample = e(mean_sample)
			local mean_disp = e(mean_disp)		
			local N = e(N)
			matrix b = e(b) 
			matrix b = b[1,1..7]
			matrix V = e(V)
			matrix V = V[1..7,1..7]
			local obs : disp %9.0f `N'
			quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
				estadd local mean_sample   " `mean_sample' "
				estadd local mean_disp   " `mean_disp' "			
				estadd local obs  " `obs' "															
		eststo `fe'fe_`cov'
		
		* Women
		estimates restore `fe'fe_`cov'
		local mean_sample = e(mean_sample)
		local mean_disp = e(mean_disp)			
		nlcom		(workvolume0: 0)			///
					(workvolume1: 0 + _b[workvolume_det_4cat1] + _b[workvolume_det_4cat1Xgeschlecht]) ///
					(workvolume2: 0 + _b[workvolume_det_4cat2] + _b[workvolume_det_4cat2Xgeschlecht]) ///
					(workvolume3: 0 + _b[workvolume_det_4cat3] + _b[workvolume_det_4cat3Xgeschlecht])   , post
		estadd local mean_sample   " `mean_sample' "
		estadd local mean_disp   " `mean_disp' "					
		estimates 	store women_`fe'fe_`cov'	
		
		* Men
		estimates restore `fe'fe_`cov'
		local mean_sample = e(mean_sample)	
		local mean_disp = e(mean_disp)		
		nlcom		(workvolume0: 0)			///
					(workvolume1: _b[workvolume_det_4cat1] ) ///
					(workvolume2: _b[workvolume_det_4cat2] ) ///
					(workvolume3: _b[workvolume_det_4cat3] ) , post
		estadd local mean_sample   " `mean_sample' "	
		estadd local mean_disp   " `mean_disp' "				
		estimates 	store men_`fe'fe_`cov'
		
		* Scale estimates and standard errors of contact_button_clicked estimates: 	
		foreach est in men women {
		estimates restore `est'_`fe'fe_`cov'
				local mean_disp = e(mean_disp)
				local mean = e(mean_sample)	
				local scale: disp %9.6g  100/`mean'
				local N = e(N)
				matrix b = e(b) * `scale'
				matrix V = e(V) * `scale' ^2
				local obs : disp %8.0f `N'
				quietly ereturn post b V, dep(contact_button_clicked) obs(`obs') 	
					estadd local mean_disp   " `mean_disp' "  
					estadd local obs  " `obs' "									
		eststo `est'_`fe'fe_`cov'_sc		
		}	
		}
	}
		
	local keepvar " workvolume1 workvolume2 workvolume3 "
	coefplot 	(men_stesfe_lasso_sc, 		keep(workvolume0) 	nokey			color(black) 	msymbol(Oh) 				offset(+0.1)) ///
				(men_stesfe_lasso_sc, 		keep(`keepvar') 	label(Men)		mcolor(black)	ciopts(lcolor(black)) 	offset(+0.1)) ///
				(women_stesfe_lasso_sc,  		keep(workvolume0 ) 	nokey			color(gs10) 		msymbol(Oh) 				offset(-0.1)) ///
				(women_stesfe_lasso_sc,  		keep(`keepvar') 	label(Women)	mcolor(gs10) 	ciopts(lcolor(gs10)) 	offset(-0.1)) , ///
				graphregion(color(white)) bgcolor(white) xline(0,lcolor(black) lpattern(dash)) ///
				order(workvolume0 workvolume1 workvolume2 workvolume3 ) ///
				coeflabels(workvolume0 = "Full-time (Ref.-Cat.)" workvolume1 = "80%-99%" workvolume2 = "60%-79%" workvolume3 = "<60%") ///
				xtitle("Effect on prob. that contact button is clicked (in %)") 	
	graph export "$results_part_time\figure_3.eps", as(eps) replace
	
	* Produce table to have numerical estimates
	local keepvar " workvolume1 workvolume2 workvolume3 "
	esttab men_stesfe_lasso_sc women_stesfe_lasso_sc , ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01)  ///
		keep( `keepvar') ///
		refcat(workvolume1 "Full-time (Ref.-Cat.)" ) ///
		coeflabels( workvolume1  "80%-99%" workvolume2  "60%-79%" workvolume3  "<60%") ///
		order(workvolume0 workvolume1 workvolume2 workvolume3 ) ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
	esttab men_gender_sc women_gender_sc using "$results_part_time\figure_3_table.csv", replace ///  
		varwidth(30) cells("b(star fmt(%9.3gc)) se(par fmt(%6.2gc))") label nodepvar collabels(none) nomtitles star(* 0.1 ** 0.05 *** 0.01) ///
		keep( `keepvar') ///
		refcat(workvolume1 "Full-time (Ref.-Cat.)" ) ///
		coeflabels( workvolume1  "80%-99%" workvolume2  "60%-79%" workvolume3  "<60%") ///
		order(workvolume0 workvolume1 workvolume2 workvolume3 ) ///
		title(Effect of desired workload on probability of a contact attempt by gender) ///
		stats(mean_disp  N , label("Mean dep. Var." "Observations") fmt(%12.3gc) ) ///
		addnotes("All Regressions include Search fixed effects" "Standard Errors clustered at user level")		
